

# Neutron diffusion

Graham W. Griffiths  
City, University of London  
oo0oo

## Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Eigenfunctions</b>	<b>4</b>
<b>3</b>	<b>Nuclear reaction - critical mass</b>	<b>4</b>
3.1	1D case in Cartesian coordinates . . . . .	5
3.1.1	Calculate the coefficients $a_p$ . . . . .	7
3.1.2	Maple program for 1D case . . . . .	7
3.2	2D case in Cartesian coordinates . . . . .	7
3.2.1	Calculate the coefficients $a_{p,q}$ . . . . .	9
3.2.2	Maple program for 2D case . . . . .	9
3.3	3D case in Cartesian coordinates . . . . .	11
3.3.1	Calculate the coefficients $a_{p,q,r}$ . . . . .	12
3.3.2	Maple program for 3D case - Cartesian coordinates . . . . .	13
3.4	3D case in cylindrical coordinates . . . . .	14
3.4.1	Calculate the coefficients, $a_{p,q}$ . . . . .	17
3.4.2	Maple program for 3D case - cylindrical coordinates . . . . .	17
3.5	3D case in spherical coordinates - Dirichlet boundary conditions . . . . .	19
3.5.1	Calculate the coefficients, $a_p$ . . . . .	21
3.5.2	Maple program for 3D case - spherical coordinates and Dirichlet BC . . . . .	22
3.6	3D case in spherical coordinates - Neumann boundary condition . . . . .	24
3.6.1	Maple program for 3D case - spherical coordinates and Neumann BC . . . . .	25
<b>4</b>	<b>Determining eigenfunction coefficients</b>	<b>27</b>
<b>5</b>	<b>The addition of a tamper</b>	<b>28</b>
<b>6</b>	<b>Neutron mean free path</b>	<b>29</b>
<b>A</b>	<b>Orthogonality of Bessel functions</b>	<b>31</b>
<b>B</b>	<b>Bessel function zeros</b>	<b>33</b>
<b>C</b>	<b>Physical parameters for <math>U^{235}</math> and <math>Pu^{239}</math></b>	<b>33</b>
<b>D</b>	<b>Summary of critical conditions</b>	<b>33</b>

## 1 Introduction

The *Manhattan Project* was located at Los Alamos, New Mexico, US, and culminated in the production of the first nuclear weapons. It was started in 1939, following a letter outlining concerns that Germany may be developing nuclear bomb technology, signed by Albert Einstein and sent to President Franklin D. Roosevelt. The project was led by the United States with the support of the United Kingdom and Canada. In 1942 one of the leading American theoretical physicists, J. Robert Oppenheimer, became the director of the Los Alamos National Laboratory. He was supported by a team drawn from many countries, of the foremost scientists and engineers at the time. The project culminated

---

<sup>1</sup>email: [graham@griffiths1.com](mailto:graham@griffiths1.com)




**Figure 1:** Photograph of the *Trinity* explosion, taken by Jack W. Aeby, 16th July, 1945. Aeby was a civilian worker at Los Alamos laboratory, supervised by the Manhattan Project organization. Source: originally from *Life Magazine*, now in the public domain.

on 16th July 1945, with the first nuclear explosion *Trinity*, quickly followed by nuclear bombs being dropped on Hiroshima and Nagasaki, nicknamed *Little Boy* and *Fat Man* respectively - see Figs. (1), (2), (3) and (4). This abruptly ended the war with Japan. The definitive history of the Manhattan Project and development of the first atomic bomb is available in the excellent book by Richard Rhodes: *The Making of the Atomic Bomb* [Rho-86], for which he won the 1988 Pulitzer Prize for General Non-fiction. A 25th anniversary edition of this book was published in 2012.


In this paper we are concerned with the diffusion of neutrons in fissile material where collisions between free neutrons and nuclei result in the release of secondary neutrons. As the fissile material increases in size, and depending upon the mean free path for neutron travel, the radioactive material will become critical when the total density of neutrons increases exponentially. The result is a runaway nuclear reaction that can lead to an intense explosion. The calculations which follow establish the size at which criticality occurs. These calculations were central to the development of a successful nuclear bomb, and were first derived as part of the Manhattan Project. Our calculations are mainly based upon simplified Dirichlet boundary conditions, whereby the neutron density is assumed to fall to zero at the edges of the core, i.e. no neutrons escape. This is more restrictive than the true Neumann boundary conditions, where neutrons do actually escape, and leads to a slightly larger core size estimate, but well within the ball park of what happens in practice. For the spherical coordinates problem, we also include calculations using Neumann boundary conditions.



**Figure 2:** Unidentified person with the Trinity device, code named *the Gadget*, shortly before the test at Alamogordo, New Mexico. Source: National Security Archive, George Washington University.



**Figure 3:** A nuclear weapon of the "Little Boy" type, the uranium gun-type detonated over Hiroshima. It is 28 inches in diameter and 120 inches long. "Little Boy" weighed about 9,000 pounds and had a yield approximating 15,000 tons of high explosives. Source: U.S. National Archives, RG 77-AEC



**Figure 4:** A nuclear weapon of the "Fat Man" type, the plutonium implosion type detonated over Nagasaki. 60 inches in diameter and 128 inches long, the weapon weighed about 10,000 pounds and had a yield approximating 21,000 tons of high explosives Source: U.S. National Archives, RG 77-AEC.

Maple computer programs are included which perform diffusion simulations and produce neutron density plots for different diffusion scenarios.

## 2 Eigenfunctions

Recall from the study of differential equations that the solutions to:

$$\frac{dw}{dx} = kw \quad \text{and} \quad D \frac{d^2w}{dx^2} = kw, \quad w = w(x), \quad x \in \mathbb{R}, \quad (1)$$

are

$$w = A_1 \exp(kt) \quad \text{and} \quad A_2 \exp(\pm \sqrt{k/D} x), \quad (2)$$

respectively; with constants  $A_1$  and  $A_2$  determined by the initial conditions (ICs). These results are then used to solve the diffusion equation by the *variable separable* method.

Given the following partial differential equation (PDE) that describes a diffusion process,

$$\frac{\partial u}{\partial t} = D \frac{\partial^2 u}{\partial x^2}, \quad u = u(t, x), \quad 0 > t \geq \infty, \quad x \in \mathbb{R}, \quad (3)$$

we postulate a solution of the form

$$u(t, x) = T(t)X(x), \quad (4)$$

so that on substituting into eqn. (3) we obtain

$$\frac{\partial TX}{\partial t} = D \frac{\partial^2 TX}{\partial x^2}. \quad (5)$$

Dividing through by  $TX$  and rearranging, we obtain

$$\frac{1}{T} \frac{\partial T}{\partial t} = D \frac{1}{X} \frac{\partial^2 X}{\partial x^2} = -k, \quad (6)$$

where  $k$  is known as the *separation constant*. It is a constant because  $T$  is a function of  $t$  only, and  $X$  is a function of  $x$  only; therefore, the only solution to eqn. (5) is for  $k$  to be a constant (we explain the negative sign subsequently). Thus, we now have two differential equations to solve, namely

$$\frac{dT}{dt} = -kT, \quad \text{and} \quad \frac{d^2X}{dx^2} = -\frac{k}{D}X. \quad (7)$$

And because  $T$  and  $X$  are both only functions of a single variable, the PDE has been transformed into two ordinary differential equations (ODEs).

The solutions to eqn. (7) are those we obtained earlier for eqns. (1), that is

$$T = A \exp(-kt) \quad \text{and} \quad X = B \exp(\pm \sqrt{-k/D} x). \quad (8)$$

We can now see that without the negative sign in front of  $k$  in eqn. (6), the above solutions for  $T$  and  $X$  would be physically unrealistic. This is because they would become unbounded as  $t \rightarrow \infty$  and  $x \rightarrow \infty$ .

In eqns. (8),  $-k$  and  $\sqrt{-k/D}$  ( $= i\sqrt{k/D}$ ) represent the solution *eigenvalues* for  $T$  and  $X$  respectively, and the corresponding right hand sides of the solutions are known as *eigenfunctions*.

## 3 Nuclear reaction - critical mass

Consider the diffusion of uranium 235 ( $\text{U}^{235}$ ) neutrons which can be described mathematically on domain  $\Omega$  with boundary  $\partial\Omega$ , by the following diffusion equation that includes

a source

$$\frac{\partial n}{\partial t} = \mu \nabla^2 n + \eta n, \quad n = n(t, x), \quad (\mu, \eta) > 0, \quad 0 > t \geq \infty, \quad x \in \mathbb{R}^s, \quad s \in \{1, 2, 3\}; \quad (9)$$

$$\text{BCs: } n(t, 0) = 0, \quad n(t, L) = 0; \quad \text{IC: } n(0, x) = f(x);$$

where  $n$  represents neutron density,  $\mu$  the diffusion constant [ $\text{m}^2/\text{s}$ ],  $\eta$  the neutron rate of formation. Here we assume that neutrons at the boundaries simply escape and disappear, so the boundary conditions (BCs) are of the Dirichlet type, and are set to zero. However, a more realistic BC is of the Neumann type for which a solution is calculated in Section (3.6). The initial condition (IC) is given by  $n(0, x) = f(x)$ . The diffusion constant is equal to  $\mu = \lambda_t v_{neut}/3$ , where  $\lambda_t$  represents neutron *transport free path*, and  $v_{neut}$  represents *neutron speed*. The neutron rate of formation constant is equal to  $\eta = v_{neut}(\nu - 1)/\lambda_f$ , where  $\nu$  represents *secondary neutrons created by fission* (the  $-1$  accounts for the neutron causing fission being consumed in the reaction), and  $\lambda_f$  represents neutron *fission free path*. Using  $\text{U}^{235}$  values from Table (7) we obtain:  $\mu = 2.345\text{E}+05$  ( $\text{m}^2/\text{s}$ ) and  $\eta = 1.896\text{E}+08$  ( $1/\text{s}$ ).

The solution to eqn. (9) was first demonstrated, practically, in 1945 when the first atomic explosion was detonated in the New Mexico desert. The calculations were set out in the technical briefing that new recruits were given as an introduction to the Manhattan Project. The briefing was given as a course of five lectures by American physicist Dr Robert Serber as an *indoctrination course* during the first two weeks of April, 1943. A set of classified notes, written up by nuclear physicist Edward U Condon, was printed at Los Alamos under the title: *The Los Alamos Primer*. It was declassified in 1963 and is now in the public domain and available on line at [https://commons.wikimedia.org/wiki/File:Los\\_Alamos\\_Primer.pdf](https://commons.wikimedia.org/wiki/File:Los_Alamos_Primer.pdf). This document summarizes what was known regarding nuclear explosions around 1943. It was subsequently published as a book by University California Press [Ser-92].

### 3.1 1D case in Cartesian coordinates

We now consider the planar case on domain  $\Omega = [0, L_x]$ , which can be described mathematically by the following 1D representation of the problem described by eqn. (9)

$$\frac{\partial n}{\partial t} = \mu \frac{\partial^2 n}{\partial x^2} + \eta n, \quad n = n(t, x), \quad (\mu, \eta) > 0, \quad 0 > t \geq \infty, \quad x \in \mathbb{R};$$

$$\text{BCs: } n(t, 0) = 0, \quad n(t, L_x) = 0;$$

$$\text{IC: } n(0, x) = f(x); \quad (10)$$

where we again have Dirichlet BCs, and symbols are as described previously.

We can now use the approach of Section (2) to solve eqn. (10), where we postulate a solution of the form

$$n(t, x) = T(t)X(x), \quad (11)$$

so that on substituting into eqn. (10), this time we obtain

$$\frac{\partial TX}{\partial t} = \mu \frac{\partial^2 TX}{\partial x^2} + \eta TX. \quad (12)$$

Dividing through by  $TX$  and rearranging, we obtain

$$\frac{1}{T} \frac{\partial T}{\partial t} - \eta = \mu \frac{1}{X} \frac{\partial^2 X}{\partial x^2} = -\alpha, \quad (13)$$

where we have used the separation constant  $\alpha$  instead of  $k$ , that again has been negated. We now have two ODEs to solve, namely

$$\frac{dT}{dt} = (\eta - \alpha)T, \quad \text{and} \quad \frac{d^2X}{dx^2} = -\frac{\alpha}{\mu}X. \quad (14)$$

The solution to the first of eqns. (14) is

$$T = A \exp([\eta - \alpha]t), \quad (15)$$

from which we see immediately that for a bounded solution, we must have  $\alpha > \eta$ .

The solution to the second of eqns. (14) is

$$\begin{aligned} X &= B \exp(i\sqrt{\alpha/\mu}x), \\ &= B_1 \cos(\sqrt{\alpha/\mu}x) + B_2 \sin(\sqrt{\alpha/\mu}x), \end{aligned} \quad (16)$$

and the general solution to eqn. (9) becomes

$$n = \exp([\eta - \alpha]t) \left[ C \cos(\sqrt{\alpha/\mu}x) + D \sin(\sqrt{\alpha/\mu}x) \right], \quad (17)$$

where we have combined constants. From the boundary conditions  $n$  must be zero at  $x = 0$  and  $x = L$  which implies that we must have  $C = 0$ . It also means that these requirements will be met when the argument of the sin term is equal to  $p\pi x/L$ , where  $p \in \mathbb{Z}$ . But, eqn. (9) is linear and therefore the *superposition principal* applies, and the solution becomes

$$n = \sum_{p=1}^{\infty} a_p \exp([\eta - \alpha]t) \sin\left(\frac{p\pi}{L}x\right), \quad (18)$$

where constants  $a_p (= D)$  are determined by the initial and/or boundary conditions. Note that we could have taken the summation from  $-\infty$  to  $+\infty$ , but this is not necessary - see section (4).

For eqn. (18) to be correct, we must have the relationship

$$\sqrt{\frac{\alpha}{\mu}} = \frac{p\pi}{L} \quad \Rightarrow \quad \alpha = \mu \left(\frac{p\pi}{L}\right)^2, \quad (19)$$

for  $p = 0, 1, 2, \dots$ . Hence, eqn. (18) becomes

$$n = \sum_{p=1}^{\infty} a_p \exp\left(\eta t - \mu \frac{p^2\pi^2}{L^2}t\right) \sin\left(\frac{p\pi}{L}x\right). \quad (20)$$

Therefore, it follows directly that for  $n$  to increase unbounded, the *critical condition*,

$$\mu \frac{p^2\pi^2}{L^2} < \eta, \quad \text{or} \quad L > p\pi \sqrt{\frac{\mu}{\eta}} \quad (21)$$

for all  $p$ . However, the worst case (shortest  $L$ ) is for  $p = 1$ , when the critical length is given by

$$L_{\text{crit}} = \pi \sqrt{\frac{\mu}{\eta}}. \quad (22)$$

Using values for  $U^{235}$  from Table (7), we obtain the critical length for the 1D case

$$L_{\text{crit}} = 11.05 \text{ cm}. \quad (23)$$

This, of course, represents a non-physical situation.



### 3.1.1 Calculate the coefficients $a_p$

Calculations for the coefficients  $a_p$  are included in Section (4), where we obtain

$$a_p = \frac{2}{L} \int_0^L f(x) \sin\left(\frac{p\pi}{L}x\right) dx, \quad p = 1, 2, \dots, \quad (24)$$

and values for  $a_p$  are given in Table (5).

### 3.1.2 Maple program for 1D case

A Maple program that calculates the coefficients  $a_p$  of Table (5), and simulates eqn. (18) with parameters from Table(7), is given in Listing (1). The corresponding solution from  $t = 0$  to  $t = 1\text{E-}07$  is provided in Figure (5) as a surface plot. Increasing the number of terms beyond 30 makes negligible difference to the result.

**Listing 1:** Maple program to calculate coefficients  $a_p$  and simulate eqn. (20) .

---

```
restart;with(Physics):
# Gaussian initial condition
f(x):=A*exp(-lambda*((x-L/2)/(L/2))^2);
L0:=0.111;L:=L0;A:=1;lambda:=100;
plot(f(x),x=0..L,title="f(x)"); unassign('L');
#Calculate symbolic solution for a[p]
eqn3:=(2/L)*Int(f(x)*sin(p*Pi*x/L),x=0..L) =
(2/L)*int(f(x)*sin(p*Pi*x/L),x=0..L) assuming(p,integer, 1,integer);
unassign('p');
# Calculate values for a[p]
N:=30;L:=L0;lambda:=100;
mu:=2.3446e+05;eta:=1.8958e+8;
Lcrit := evalf(Pi*sqrt(mu/eta));
for p from 1 by 1 to N do
aa[p]:=evalf(rhs(eqn3)); #print(aa[p]);
end do;
# Calculate solution
n:= sum(aa[i]*exp(eta*t-mu*(i*Pi/L)^2*t)*sin(i*Pi*x/L),i=1..N):
# Plot solution surface
plot3d(n,x=0..L,t=0..2e-5,axes=framed,title="Neutron diffusion,
L=11.1cm, N=30",labels=["x", "t", "n(t,x)"],orientation=[-48, 69, 1]);
# Calculate errors at t=0
plot(subs(t=0,n)-f(x),x=0..L,axes=framed,title="Neutron diffusion,
L=3.5, N=20, t=0",labels=["x","n(x,t=0)"],thickness=2);
```

---

## 3.2 2D case in Cartesian coordinates


We now consider the 2D case on domain  $\Omega = [0, L_x] \times [0, L_y]$ , which can be described mathematically by the following representation of the problem described by eqn. (9)

$$\begin{aligned} \frac{\partial n}{\partial t} &= \mu \left( \frac{\partial^2 n}{\partial x^2} + \frac{\partial^2 n}{\partial y^2} \right) + \eta n, \quad n = n(t, x, y), \quad (\mu, \eta) > 0, \quad 0 > t \geq \infty, \quad (x, y) \in \mathbb{R}; \\ \text{BCs: } n(t, 0, y) &= 0, \quad n(t, x, 0) = 0, \quad n(t, L_x, y) = 0, \quad n(t, x, L_y) = 0; \\ \text{IC: } n(0, x, y) &= f(x, y); \end{aligned} \quad (25)$$

where we again have zero Dirichlet BCs, and symbols are as described previously.

We use the approach of Section (2) to solve eqn. (40), where this time we postulate a solution of the form

$$n(t, x, y) = T(t)X(x)Y(y), \quad (26)$$



**Figure 5:** Plot of  $n(t, x)$  .

so that on substituting into eqn. (40), we obtain

$$\frac{\partial TXY}{\partial t} = \mu \frac{\partial^2 TXY}{\partial x^2} + \mu \frac{\partial^2 TXY}{\partial y^2} + \eta TXY. \quad (27)$$

Dividing through by  $TXY$  and rearranging, we obtain

$$\frac{1}{T} \frac{\partial T}{\partial t} - \eta = \mu \frac{1}{X} \frac{\partial^2 X}{\partial x^2} + \mu \frac{1}{Y} \frac{\partial^2 Y}{\partial y^2} = -\alpha, \quad (28)$$

where we again use the negated separation constant  $\alpha$  on the right hand side.

We now have three ODEs to solve, namely

$$\frac{dT}{dt} = (\eta - \alpha)T, \quad \frac{d^2 X}{dx^2} = -\frac{\alpha_1}{\mu} X \quad \text{and} \quad \frac{d^2 Y}{dy^2} = -\frac{\alpha_2}{\mu} Y, \quad (29)$$

where we have split the constant  $\alpha$  for the spatial derivatives, such that  $\alpha = \alpha_1 + \alpha_2$ .

Following the same arguments as used for the 1D case, the solution is found to be

$$n = \sum_{p,q} a_{p,q} \exp([\eta - \alpha]t) \sin\left(\frac{p\pi}{L_x}x\right) \sin\left(\frac{q\pi}{L_y}y\right), \quad (30)$$

where constants  $a_{p,q}$  are determined by the initial and/or boundary conditions.

For eqn. (30) to be correct, the following relationships must hold

$$\begin{aligned} \sqrt{\frac{\alpha_1}{\mu}} &= \frac{p\pi}{L_x} &\Rightarrow &\alpha_1 = \mu \left(\frac{p\pi}{L_x}\right)^2, \\ \sqrt{\frac{\alpha_2}{\mu}} &= \frac{q\pi}{L_y} &\Rightarrow &\alpha_2 = \mu \left(\frac{q\pi}{L_y}\right)^2, \end{aligned} \quad (31)$$



for  $p, q = 0, 1, 2, \dots$ . Hence, eqn. (30) becomes

$$n = \sum_{p,q}^{\infty} a_{p,q} \exp \left( \eta t - \mu \pi^2 \left[ \frac{p^2}{L_x^2} + \frac{q^2}{L_y^2} \right] t \right) \sin \left( \frac{p\pi}{L_x} x \right) \sin \left( \frac{q\pi}{L_y} y \right), \quad (32)$$

Therefore, it follows directly that for  $n$  to increase unbounded, the critical condition,

$$\mu \pi^2 \left[ \frac{p^2}{L_x^2} + \frac{q^2}{L_y^2} \right] < \eta, \quad (33)$$

must apply for all  $p, q$ . However, the worst case (shortest  $L$ ) is for  $p = 1$  and  $q = 1$ , when the critical condition is given by

$$\left[ \frac{1}{L_x^2} + \frac{1}{L_y^2} \right]^{\frac{1}{2}} = \frac{1}{\pi} \sqrt{\frac{\eta}{\mu}}. \quad (34)$$

Now, if the domain is symmetric, that is  $L_x = L_y = L$ , we get the critical dimension for a square domain,  $\Omega$

$$L_{\text{crit}} = \pi \sqrt{\frac{2\mu}{\eta}}. \quad (35)$$

Using values for  $U^{235}$  from Table (7), we obtain the critical length for the 2D case

$$L_{\text{crit}} = 15.62 \text{ cm}. \quad (36)$$

This again represents a non-physical situation.

### 3.2.1 Calculate the coefficients $a_{p,q}$

To calculate the coefficients  $a_{p,q}$  we apply the ideas of Section (4) when we obtain

$$a_{p,q} = \frac{4}{L^2} \int_0^L \int_0^L f(x, y) \sin \left( \frac{p\pi}{L} x \right) \sin \left( \frac{q\pi}{L} y \right) dx dy, \quad p = 1, 2, \dots \quad q = 1, 2, \dots \quad (37)$$

If we set the initial condition to

$$f(x, y, z) = \frac{16xy}{L^2} \left( 1 - \frac{x}{L} \right) \left( 1 - \frac{y}{L} \right), \quad (38)$$

which satisfies the boundary conditions, and solve using Maple, the coefficients are found to be

$$a_{p,q} = 256 \frac{1 - (-1)^q - (-1)^p + (-1)^{p+q}}{p^3 q^3 \pi^6}. \quad (39)$$


Substituting  $p, q$  from 1 to 5 in eqn. (39), we obtain the 25 coefficients shown in Table (1).

### 3.2.2 Maple program for 2D case

A Maple program that calculates the coefficients  $a_p$  and simulates eqn. (32) with parameters from Table (7), is given in Listing (2). The corresponding solution at  $t=1\text{e-}7$  [s] is provided in Figure (6) as a surface plot. Increasing the number of terms beyond 20 makes negligible difference to the result.

**Table 1:** Values for coefficients  $a_{p,q}$  calculated from eqn. (39).

$p$	$q$	$a_{p,q}$	$p$	$q$	$a_{p,q}$
1	1	1.0651E+00	3	4	0.0000E+00
1	2	0.0000E+00	3	5	3.1559E-04
1	3	3.9449E-02	4	1	0.0000E+00
1	4	0.0000E+00	4	2	0.0000E+00
1	5	8.5210E-03	4	3	0.0000E+00
2	1	0.0000E+00	4	4	0.0000E+00
2	2	0.0000E+00	4	5	0.0000E+00
2	3	0.0000E+00	5	1	8.5210E-03
2	4	0.0000E+00	5	2	0.0000E+00
2	5	0.0000E+00	5	3	3.1559E-04
3	1	3.9449E-02	5	4	0.0000E+00
3	2	0.0000E+00	5	5	6.8168E-05
3	3	1.4611E-03			



**Figure 6:** Plot of  $n(t=1e-7, x, y)$  .

**Listing 2:** Maple program to calculate coefficients  $a_{p,q}$  and simulate eqn. (32) .

```

restart;with(Physics):wilt(plots):
# Neutron diffusion in Cartesian coordinates
# Initial condition
# f(x,y):=A*exp(-xi*((x-L/2)^2+(y-L/2)^2)/(L/2+L/2)^2); # Gaussian
f(x,y):=(1-(x/L))*(1-(y/L))*x*y/(L/4)^2;
L0:=15.7;L:=L0;A:=1;xi:=100;
plot3d(f(x,y),x=0..L,y=0..L,title="Neutron diffusion",
n=f(x,y), t=0",axes="normal",labels=["x", "y", "n(t=0,x,y)"],orientation=[-48,
69, 1]);
unassign('L');
#Calculate symbolic solution for a[p]
eqn3:=(4/L^2)*Int(Int(f(x,y)*sin(p*Pi*x/L)*sin(q*Pi*y/L),x=0..L),y=0..L) =
(4/L^2)*int(int(f(x,y)*sin(p*Pi*x/L)*sin(q*Pi*y/L),x=0..L),y=0..L)
assuming(p, integer, q, integer);
unassign('p','q');
# Calculate values for a[p,q]
N:=5;L:=L0;lambda:=100;
mu:=2.3446e5;eta:=1.8958e8;
for p from 1 by 1 to N do

```

```

for q from 1 by 1 to N do
    aa[p,q]:=evalf(rhs(eqn3)); # print(p,q,aa[p,q]);
end do;
end do:
# Calculate solution
n:= sum(sum(aa[i,j]*exp(eta*t-mu*((i*Pi/L)^2+(j*Pi/L)^2)*t)*
        sin(i*Pi*x/L)*sin(j*Pi*y/L),i=1..N),j=1..N):
# Plot solution surface
t:=1e-7;
plot3d(n,x=0..L,y=0..L,axes=framed,title="Neutron diffusion",
L=15.7cm, N=20, t=1e-7,labels=["x", "y", "n(t=1e-7,x,y)"],orientation=[-48,
69, 1]);
t:=0;
plot3d(n-f(x,y),x=0..L,y=0..L,axes=framed,title="Neutron diffusion error",
L=15.7cm, N=20, t=0,labels=["x", "y", "error(0,x,y)"],orientation=[-48, 69,
1]);

```

---

### 3.3 3D case in Cartesian coordinates

We now consider the 3D case on domain  $\Omega = [0, L_x] \times [0, L_y] \times [0, L_z]$ , which can be described mathematically by the following representation of the problem described by eqn. (9)

$$\begin{aligned}
 \frac{\partial n}{\partial t} &= \mu \left( \frac{\partial^2 n}{\partial x^2} + \frac{\partial^2 n}{\partial y^2} + \frac{\partial^2 n}{\partial z^2} \right) + \eta n, \quad n = n(t, x, y, z), \\
 (\mu, \eta) &> 0, \quad 0 > t \geq \infty, \quad (x, y, z) \in \mathbb{R}; \\
 \text{BCs:} \quad &n(t, 0, y, z) = 0, \quad n(t, x, 0, z) = 0, \quad n(t, x, y, 0) = 0, \\
 &n(t, x, L_y, L_z) = 0, \quad n(t, L_x, y, L_z) = 0, \quad n(t, L_x, L_y, z) = 0; \\
 \text{IC:} \quad &n(0, x, y, z) = f(x, y, z);
 \end{aligned} \tag{40}$$

where we again have zero Dirichlet BCs, and symbols are as described previously.

Applying previous arguments to the 3D case, and defining  $\alpha = \alpha_1 + \alpha_2 + \alpha_3$ , we obtain

$$n = \sum_{p,q,r}^{\infty} a_{p,q,r} \exp \left( \eta t - \mu \pi^2 \left[ \frac{p^2}{L_x^2} + \frac{q^2}{L_y^2} + \frac{r^2}{L_z^2} \right] t \right) \sin \left( \frac{p\pi}{L_x} x \right) \sin \left( \frac{q\pi}{L_y} y \right) \sin \left( \frac{r\pi}{L_z} z \right), \tag{41}$$

and the worst case (shortest  $L$ ) is for  $p = 1, q = 1$  and  $r = 1$ , when the critical condition is given by

$$\left[ \frac{1}{L_x^2} + \frac{1}{L_y^2} + \frac{1}{L_z^2} \right]^{\frac{1}{2}} = \frac{1}{\pi} \sqrt{\frac{\eta}{\mu}}. \tag{42}$$

Now, if the domain is symmetric, that is  $L_x = L_y = L_z = L$ , we get the critical dimension for a 3D domain,  $\Omega$ , having the shape of a cube,

$$L_{\text{crit}} = \pi \sqrt{\frac{3\mu}{\eta}}. \tag{43}$$

Using values for  $U^{235}$  from Table (7), we obtain the critical length and volume for the 3D Cartesian coordinates case

$$L_{\text{crit}} = 19.14 \text{ cm}, \quad V_{\text{crit}} = 7007 \text{ cm}^3. \tag{44}$$

This is a realizable situation and corresponds to a mass of around 131 kg.

### 3.3.1 Calculate the coefficients $a_{p,q,r}$

To calculate the coefficients  $a_{p,q}$  we apply the ideas of Section (4) when we obtain

$$a_{p,q,r} = \frac{8}{L^3} \int_0^L \int_0^L \int_0^L f(x, y, z) \sin\left(\frac{p\pi}{L}x\right) \sin\left(\frac{q\pi}{L}y\right) \sin\left(\frac{r\pi}{L}z\right) dx dy dz, \quad (45)$$

$$p = 1, 2, \dots \quad q = 1, 2, \dots \quad r = 1, 2, \dots$$

If we set the initial condition to

$$f(x, y, z) = \frac{8xyz}{L^3} \left(1 - \frac{x}{L}\right) \left(1 - \frac{y}{L}\right) \left(1 - \frac{z}{L}\right), \quad (46)$$

which satisfies the boundary conditions, and solve using Maple, the coefficients are found to be

$$a_{p,q,r} = -512 \frac{(-1)^q - (-1)^{q+r} - 1 + (-1)^r + (-1)^p - (-1)^{p+r} - (-1)^{p+q} + (-1)^{p+q+r}}{p^3 q^3 r^3 \pi^9}. \quad (47)$$


Substituting  $p, q, r$  from 1 to 4 in eqn. (47), we obtain the 64 coefficients shown in Table (2). Due to the numerator containing minus ones raised to integer powers, this results in many zero coefficients.

**Table 2:** Values for coefficients  $a_{p,q,r}$  calculated from eqn. (47). Note the large number of zero coefficients.

$p$	$q$	$r$	$a_{p,q,r}$	$p$	$q$	$r$	$a_{p,q,r}$
1	1	1	1.3741E-01	3	1	1	5.0892E-03
1	1	2	0.0000E+00	3	1	2	0.0000E+00
1	1	3	5.0892E-03	3	1	3	1.8849E-04
1	1	4	0.0000E+00	3	1	4	0.0000E+00
1	2	1	0.0000E+00	3	2	1	0.0000E+00
1	2	2	0.0000E+00	3	2	2	0.0000E+00
1	2	3	0.0000E+00	3	2	3	0.0000E+00
1	2	4	0.0000E+00	3	2	4	0.0000E+00
1	3	1	5.0892E-03	3	3	1	1.8849E-04
1	3	2	0.0000E+00	3	3	2	0.0000E+00
1	3	3	1.8849E-04	3	3	3	6.9810E-06
1	3	4	0.0000E+00	3	3	4	0.0000E+00
1	4	1	0.0000E+00	3	4	1	0.0000E+00
1	4	2	0.0000E+00	3	4	2	0.0000E+00
1	4	3	0.0000E+00	3	4	3	0.0000E+00
1	4	4	0.0000E+00	3	4	4	0.0000E+00
2	1	1	0.0000E+00	4	1	1	0.0000E+00
2	1	2	0.0000E+00	4	1	2	0.0000E+00
2	1	3	0.0000E+00	4	1	3	0.0000E+00
2	1	4	0.0000E+00	4	1	4	0.0000E+00
2	2	1	0.0000E+00	4	2	1	0.0000E+00
2	2	2	0.0000E+00	4	2	2	0.0000E+00
2	2	3	0.0000E+00	4	2	3	0.0000E+00
2	2	4	0.0000E+00	4	2	4	0.0000E+00
2	3	1	0.0000E+00	4	3	1	0.0000E+00
2	3	2	0.0000E+00	4	3	2	0.0000E+00
2	3	3	0.0000E+00	4	3	3	0.0000E+00
2	3	4	0.0000E+00	4	3	4	0.0000E+00
2	4	1	0.0000E+00	4	4	1	0.0000E+00
2	4	2	0.0000E+00	4	4	2	0.0000E+00
2	4	3	0.0000E+00	4	4	3	0.0000E+00
2	4	4	0.0000E+00	4	4	4	0.0000E+00

### 3.3.2 Maple program for 3D case - Cartesian coordinates

A Maple program that calculates the coefficients,  $a_{p,q,r}$ , of Table (2), using parameters from Table (7), and also simulates eqn. (41), is given in Listing (3). A corresponding plot at time  $t=1e-7$  [s] is provided in Figure (7) as a  $x$ - $y$  plane slice through the solution at  $z = L/2$ . Increasing the number of terms beyond 4 makes negligible difference to the result.



**Figure 7:** 3D neutron diffusion solution in Cartesian coordinates. Plot of a slice through the  $x$ - $y$  plane at  $z = L/2$  and time  $t = 2E-7$ .

**Listing 3:** Maple program to calculate coefficients  $a_p$  and simulate eqn. (96) .

---

```

restart;with(Physics):with(plots):
# 3D Neutron diffusion in Cartesian coordinates
# Initial condition
#f(x,y,z):=A*exp(-xi*((x-L/2)^2+(y-L/2)^2+(z-L/2)^2)/(L/2+L/2+L/2)^2); #
    Gaussian
f(x,y,z):=(1-(x/L))*(1-(y/L))*(1-(z/L))*x*y*z/(L/2)^3;
f2:=value(subs(z=L/2,f(x,y,z)));
L0:=19.2/100;L:=L0;A:=1;xi:=100;
unassign('L');
#Calculate symbolic solution for a[p]
eqn3:=(8/L^3)*Int(Int(Int(f(x,y,z)*sin(p*Pi*x/L)*sin(q*Pi*y/L)*sin(r*Pi*z/L),x
    =0..L),y=0..L),z=0..L) =
    simplify((8/L^3)*int(int(int(f(x,y,z)*sin(p*Pi*x/L)*sin(q*Pi*y/L)*sin(r*Pi*z/
        L),x=0..L),y=0..L),z=0..L),size)
    assuming(p,integer, q,integer, r,integer);
unassign('p','q','r');
# Calculate values for a[p,q,r]
N:=4;L:=L0;
# mu:=1;eta:=1;
mu:=2.3446e5;eta:=1.8958e8;
Lcrit:=evalf(Pi*sqrt(3*mu/eta));
for p from 1 by 1 to N do
    for q from 1 by 1 to N do
        for r from 1 by 1 to N do
            aa[p,q,r]:=evalf(rhs(eqn3)); # print(p,q,r,aa[p,q,r]);
        end do;
    end do;
end do;
# Calculate solution

```

```

n:= sum(sum(sum(aa[i,j,k]*exp(eta*t-mu*Pi*((i/L)^2+(j/L)^2+(k/L)^2)*t)*
    sin(i*Pi*x/L)*sin(j*Pi*y/L)*sin(k*Pi*z/L),i=1..N),j=1..N),k=1..N):
# Plot solution surface
f(x,y,z);
L:=L0;
z:=L/2;
t:=2e-7;
#f2:=value(subs(z=L/2,f(x,y,z)));
plot3d(n,x=0..L,y=0..L,axes=framed,title="3D Neutron diffusion - cube,
L=19.2cm, N=4, t=2e-7",labels=["x", "y", "n(t=2e-7,x,y,L/2)"],orientation=[-48,
    69, 1]);
t:=0;
plot3d(f2,x=0..L,y=0..L,axes=framed,title="Neutron diffusion,
L=19.2cm, N=4, t=0",labels=["x", "y", "f(x,y,L/2)"],orientation=[-48, 69, 1]);
densityplot(f2,x=0..L,y=0..L,grid=[125, 125],style=PATCHNOGRID,color=RED);

```

---

### 3.4 3D case in cylindrical coordinates

We consider now neutron diffusion in a 3D symmetrical situation. The Laplacian in cylindrical coordinates for  $n = n(r, \phi, z)$ , is given by

$$\nabla^2 n = \left( \frac{\partial^2}{\partial r^2} + \frac{1}{r} \frac{\partial}{\partial r} \right) n + \frac{1}{r^2} \frac{\partial^2}{\partial \phi^2} n + \frac{\partial^2}{\partial z^2} n. \quad (48)$$

However, for this case, we have radial symmetry and the partial derivative terms involving  $\phi$  are equal to zero. Equation (48) therefore reduces to

$$\nabla^2 n = \left( \frac{\partial^2}{\partial r^2} + \frac{1}{r} \frac{\partial}{\partial r} \right) n + \frac{\partial^2}{\partial z^2} n. \quad (49)$$

and the neutron diffusion equation with zero Dirichlet boundary conditions becomes

$$\begin{aligned} \frac{\partial n}{\partial t} &= \mu \left( \frac{\partial^2}{\partial r^2} + \frac{1}{r} \frac{\partial}{\partial r} \right) n + \mu \frac{\partial^2}{\partial z^2} n + \eta n, \\ n &= n(t, r, z), \quad (\mu, \eta) > 0, \quad 0 > t \geq \infty, \quad r \in \mathbb{R}; \\ \text{BCs: } n(t, r_1, z) &= 0, \quad n(t, r, 0) = 0, \quad n(t, r, L) = 0; \\ \text{IC: } n(0, r, z) &= f(r, z); \end{aligned} \quad (50)$$

where  $r$  represents the radial distance from the center of the cylinder,  $L$  the length of the cylinder and  $z$  the vertical distance from the base of the cylinder, other symbols are as previously defined.

We use the approach of Section (2) to solve eqn. (50), where this time we postulate a solution of the form

$$n(t, r) = T(t)R(r)Z(z), \quad (51)$$

so that on substituting into eqn. (50), we obtain

$$\frac{\partial TRZ}{\partial t} = \mu \frac{\partial^2 TRZ}{\partial r^2} + \mu \frac{1}{r} \frac{\partial TRZ}{\partial r} + \mu \frac{\partial TRZ}{\partial z^2} + \eta TRZ. \quad (52)$$

Dividing through by  $TRZ$  and rearranging, we obtain

$$\frac{1}{T} \frac{\partial T}{\partial t} - \eta = \mu \frac{1}{R} \frac{\partial^2 R}{\partial r^2} + \mu \frac{1}{rR} \frac{\partial R}{\partial r} + \mu \frac{1}{Z} \frac{\partial^2 Z}{\partial z^2} = -\ell, \quad (53)$$

where we use the negated separation constant  $\ell$  on the right hand side.



We now have three ODEs to solve, and the first is

$$\frac{dT}{dt} = (\eta - \ell)T. \quad (54)$$

This has the same solution as we obtained for the Cartesian coordinates problems, namely

$$T = A \exp([\eta - \ell]t); \quad (55)$$

and for a bounded solution, we must have  $\ell > \eta$ .

From eqn. (53) we have another variable separable problem

$$\mu \frac{1}{R} \frac{\partial^2 R}{\partial r^2} + \mu \frac{1}{rR} \frac{\partial R}{\partial r} + k = -\mu \frac{1}{Z} \frac{\partial Z}{\partial z} = k, \quad (56)$$

and the second and third ODEs become

$$\frac{dZ}{dt} = -\frac{k}{\mu}Z \quad \text{and} \quad \frac{d^2 R}{dr^2} + \frac{1}{r} \frac{dR}{dr} + \frac{\alpha - k}{\mu} R = 0. \quad (57)$$

The solution to the first of eqns. (57) is

$$Z = A \exp(i\sqrt{k/\mu}z) + B \exp(-i\sqrt{k/\mu}z). \quad (58)$$

But we require that  $Z(0) = 0$  and  $Z(L) = 0$ . The first condition requires that  $A + B = 0$  or  $B = -A$ . Thus, the solution for  $Z$  becomes

$$Z = A(\exp(i\sqrt{k/\mu}z) - \exp(-i\sqrt{k/\mu}z)) = i2A \sin(\sqrt{k/\mu}z). \quad (59)$$

The second condition when applied to eqn. (59) requires that  $\sin(\sqrt{k/\mu}L) = 0$  which can only be satisfied if the quantity  $\sqrt{k/\mu}L$  is equal to an integer multiplied by  $\pi$ . Hence, the solution for  $Z$  becomes

$$Z = i2A \sin\left(\frac{p\pi}{L}z\right), \quad p \in \mathbb{Z}. \quad (60)$$

Now consider the third ODE of ...

$$\frac{d^2 R}{dr^2} + \frac{1}{r} \frac{dR}{dr} + \frac{\ell - k}{\mu} R = 0. \quad (61)$$

On multiplying through by  $r^2$  and substituting the previously determined value for  $k$ , we obtain

$$r^2 \frac{d^2 R}{dr^2} + r \frac{dR}{dr} + r^2 \left( \frac{\ell}{\mu} - \frac{p^2 \pi^2}{L^2} \right) R = 0. \quad (62)$$

Then on making the substitution  $r = \frac{x}{\sqrt{\ell/\mu - p^2 \pi^2/L^2}}$  we obtain the standard form of Bessel's equation

$$x^2 \frac{d^2 R}{dx^2} + x \frac{dR}{dx} + x^2 R = 0. \quad (63)$$

This equation has the solution

$$R = AJ_0(x) + BY_0(x), \quad (64)$$

where  $J_0(x)$  and  $Y_0(x)$  are order 0 Bessel functions of the *first* and *second* type respectively.  $Y_0(x)$  is not finite at the origin, which we require for our solution. This means that  $B = 0$ , and on back substituting for  $x$ , the solution becomes

$$R = AJ_0 \left( r \sqrt{\frac{\ell}{\mu} - \frac{p^2 \pi^2}{L^2}} \right). \quad (65)$$

Now, we require that  $R = 0$  at  $r = r_1$ , therefore this means that eqn. (65) must be equal to

$$R = AJ_0\left(\frac{\alpha_q}{r_1}r\right) = AJ_0(\alpha_q)|_{r=r_1}, \quad (66)$$

where  $\alpha_q$  is the  $q$ th zero of  $J_0(r)$ . It therefore follows from eqns. (65) and (66) that

$$\ell = \left(\frac{\alpha_q^2}{r_1^2} + \frac{p^2\pi^2}{L^2}\right)\mu. \quad (67)$$

Inserting the solutions for  $T$ ,  $Z$  and  $R$  into eqn. (51) yields the solution

$$n(t, r, z) = J_0\left(\frac{\alpha_q}{r_1}r\right) \sin\left(\frac{p\pi z}{L}\right) \exp\left(\frac{\eta r_1^2 L^2 - \mu(\alpha_q^2 L^2 + p^2 \pi^2 r_1^2)}{r_1^2 L^2} t\right) a_{p,q}. \quad (68)$$

But, eqn. (68) is linear therefore the *superposition principal* applies, and the final solution becomes

$$n(t, r, z) = \sum_{p=1}^{\infty} \sum_{q=1}^{\infty} a_{p,q} J_0\left(\frac{\alpha_q}{r_1}r\right) \sin\left(\frac{p\pi z}{L}\right) \exp\left(\frac{\eta r_1^2 L^2 - \mu(\alpha_q^2 L^2 + p^2 \pi^2 r_1^2)}{r_1^2 L^2} t\right). \quad (69)$$

See note following eqn. (80) which shows that we can choose an appropriate initial condition  $f(r, z)$  so that  $a_{p,q} = 0$  for  $p \neq 1$ . Therefore, for this case, eqn.(69) simplifies to,

$$n(t, r, z) = \sum_{q=1}^{\infty} a_{1,q} J_0\left(\frac{\alpha_q}{r_1}r\right) \sin\left(\frac{\pi z}{L}\right) \exp\left(\frac{\eta r_1^2 L^2 - \mu(\alpha_q^2 L^2 + \pi^2 r_1^2)}{r_1^2 L^2} t\right). \quad (70)$$

Now, it follows directly from eqn. (69) that for  $n$  to increase unbounded, the *critical condition* is

$$\eta r_1^2 L^2 > \mu \alpha_q^2 L^2 + \mu p^2 \pi^2 r_1^2, \quad (71)$$

for all  $p$  and  $q$ . However, the worst case (smallest radius  $r$  and length  $L$ ) is for  $p = 1$  and  $q = 1$ , when the critical condition becomes

$$\eta r_1^2 L^2 = \mu \alpha_1^2 L^2 + \mu \pi^2 r_1^2. \quad (72)$$

Solving for radius  $r_1$

$$r_1 = \frac{\sqrt{(\eta L^2 - \pi^2 \mu) \mu} L \alpha_1}{\eta L^2 - \pi^2 \mu}, \quad (73)$$

and using this result to calculate the volume of the critical cylinder ( $\pi r_1^2 L$ ) solely in terms of  $L$ , we obtain

$$V = \frac{\pi \mu L^3 \alpha_1^2}{\eta L^2 - \pi^2 \mu}. \quad (74)$$

Differentiating  $V$  with respect to  $L$  and equating to zero, we obtain the critical length of the cylinder

$$L_{\text{crit}} = \pi \sqrt{\frac{3\mu}{\eta}}. \quad (75)$$

Substituting  $L_{\text{crit}}$  into eqn. (73) we obtain the critical radius

$$r_{\text{crit}} = \alpha_1 \sqrt{\frac{3\mu}{2\eta}}. \quad (76)$$

Using the *first* Bessel zero value from Table (6) to calculate  $\alpha_1$ , and values for  $U^{235}$  from Table (7), we obtain the critical measurements for the cylinder

$$L_{\text{crit}} = 19.14 \text{ cm}, \quad r_{\text{crit}} = 10.36 \text{ cm}, \quad V_{\text{crit}} = 6450 \text{ cm}^3. \quad (77)$$

This corresponds to a critical mass of around 121 kg.

The ratio of critical volume of a cylinder, to critical volume of a sphere, is therefore

$$\frac{V_{\text{cyl}}}{V_{\text{sphere}}} = 1.14. \quad (78)$$

All that remains now is to evaluate the coefficients  $a_{p,q}$  of eqn. (68) which will allow the implementation of a full dynamic simulation.

### 3.4.1 Calculate the coefficients, $a_{p,q}$

To calculate the coefficients  $a_{p,q}$  we need to set  $t = 0$  in eqn. (70) and follow the method outlined in Appendix (A), giving

$$a_{p,q} = \frac{4}{L r_1^2 J_1^2(\alpha_q)} \int_0^L \int_0^{r_1} J_0\left(\frac{\alpha_q}{r_1} r\right) r f(r, z) \sin\left(\frac{p\pi z}{L}\right) dr dz. \quad (79)$$

Setting the initial condition to  $f(r, z) = (1 - r^2/r_1^2) \sin(\pi z/L)$ , and rearranging the order of integration, yields

$$a_{p,q} = \frac{4}{L r_1^2 J_1^2(\alpha_q)} \int_0^{r_1} J_0\left(\frac{\alpha_q}{r_1} r\right) r \left(1 - \frac{r^2}{r_1^2}\right) \left[ \int_0^L \sin\left(\frac{\pi z}{L}\right) \sin\left(\frac{p\pi z}{L}\right) dz \right] dr. \quad (80)$$

**Note:** By inspection of eqn. (80) we observe that due to the orthogonality condition of the  $\sin()$  terms (see section A),  $a_{p,q} = 0$  for  $p \neq 1$ . Therefore, the solution simplifies to

$$a_{1,q} = \frac{4}{L r_1^2 J_1^2(\alpha_q)} \int_0^L \int_0^{r_1} J_0\left(\frac{\alpha_q}{r_1} r\right) r \left(1 - \frac{r^2}{r_1^2}\right) \sin^2\left(\frac{\pi z}{L}\right) dr dz. \quad (81)$$

Evaluating eqn. (81) using Maple, we obtain

$$a_{1,q} = \frac{4(\alpha_q J_0(\alpha_q) + 2J_1(\alpha_q))}{J_1(\alpha_q)^2 \alpha_q^3}, \quad (82)$$

which yields the coefficients in Table (3).


### 3.4.2 Maple program for 3D case - cylindrical coordinates

A Maple program that calculates the coefficients,  $a_{p,q}$ , of Table (3), using parameters from Table (7), and that also simulates eqn. (70), is given in Listing (2). The corresponding solution at time  $t=1e-5$  [s] is provided in Figure (8) as a surface plot. Increasing the number of terms beyond 30 makes negligible difference to the result.

**Table 3:** Coefficient values for 3D neutron diffusion problem in cylindrical coordinates.

$q$	$a_{1,q}$
1	1.1080E+00
2	-1.3978E-01
3	4.5476E-02
4	-2.0991E-02
5	1.1636E-02
6	-7.2212E-03
7	4.8379E-03
8	-3.4257E-03
9	2.5295E-03
10	-1.9301E-03

Neutron diffusion,  $L=19.2\text{cm}$ ,  $r_1=10.4\text{cm}$ ,  $N_p=1$ ,  $N_q=10$ ,  $t=1\text{e-}05$



**Figure 8:** Plot of  $n(r, z)$  at  $t = 1\text{E-}5$  [s] for 3D neutron diffusion, with radial symmetry, in cylindrical coordinates.

**Listing 4:** Maple program to calculate coefficients  $a_p$  and simulate eqn. (70) .

```
restart;with(plots):
digits:=30;
# Neutron diffusion in cylindrical coordinates
# Initial conditions
case:=1;
if case=1 then
    f(r,z):=A*(1-(r/r1)^2)*sin(Pi*z/L); # p=1
    p:=1;
else
    f(r,z):=A*(1-(r/r1)^2)^3*(1-(1-2*z/L)^2)^3; #sin(Pi*z/L); # p=6
end if;
L0:=0.192;L:=L0;r1:=0.104;A:=1;
plot3d(f(r,z),r=-r1..r1,z=0..L,title="f(r,z)",axes="framed",orientation=[-48,
    69, 1]);
unassign('L','r1');
#Calculate symbolic solution for a[p,q]
# Ref: Asmar, N H (2005), Partial Differential Equations: with Fourier series
# and boundary Value problems (2nd ed.). Pearson. p210 and p252
eqn3:=(2/L)*(2/(r1^2*BesselJ(1,beta[q])^2))*Int(Int(f(r,z)*BesselJ(0,beta[q]*r/
    r1)*r*sin(p*Pi*z/L),r=0..r1),z=0..L) =
    (2/L)*(2/(r1^2*BesselJ(1,beta[q])^2))*int(int(f(r,z)*BesselJ(0,beta[q]*r/r1)*
```

```

    r*sin(p*Pi*z/L),r=0..r1),z=0..L)
    assuming(p,integer,q,integer);
    unassign('p','q');
    Nq:=10;
    if case=1 then Np:=1;
    else Np:=6; end if;
    beta:=evalf(BesselJZeros(0, 1..Nq));
    # Calculate values for a[p,q]
    L0:=0.192;L:=L0;r1:=0.104;
    # mu:=1;eta:=1;
    mu:=2.3446e5;eta:=1.8958e8;
    for p from 1 by 1 to Np do
        for q from 1 by 1 to Nq do
            aa[p,q]:=evalf(rhs(eqn3)); # print(aa[p,q]);
        end do;
    end do;
    # Calculate solution
    n:= sum(sum(aa[i,j]*exp((eta*r1^2*L^2-mu*beta[j]^2*L^2-
    mu*i^2*Pi^2*r1^2)*t/(r1^2*L^2))*
    BesselJ(0, r*beta[j]/r1)*sin(i*Pi*z/L),i=1..Np),j=1..Nq):
    # Plot solution surface
    t:=1e-5;
    heading:=sprintf("Neutron diffusion, L=%4.1f, r1=%4.1f, Np=%d, Nq=%d, t=%3.2e",
        L,r1,Np,Nq,t);
    plot3d(n,r=-r1..r1,z=0..L,axes=framed,title=heading,
    labels=["r", "z", "n(r,z,t=0)"],orientation=[-48, 69, 1]);
    #plot3d(n-f(r,z),r=0..r1,z=0..L,axes=framed,title="Neutron diffusion",
    #L=4.5, Np=1, Nq=20, t=0",labels=["r", "z", "error(r,z,0)"],orientation=[-48,
    69, 1]);
    unassign('t');
    animate(plot3d,[n,r=-r1..r1,z=0..L],t=0..1e-5,frames=25,
    orientation=[-48, 69, 1],shading=zhue,labels=["r", "z", "n(r,z,t=0)"],
    style=patch,grid=[25,25],axes=framed);

```

---

### 3.5 3D case in spherical coordinates - Dirichlet boundary conditions

We consider now neutron diffusion in a 3D symmetrical situation. The Laplacian in spherical coordinates for  $n = n(r, \theta, \phi)$ , is given by

$$\nabla^2 n = \left( \frac{\partial^2}{\partial r^2} + \frac{2}{r} \frac{\partial}{\partial r} \right) n + \frac{1}{r^2 \sin \theta} \frac{\partial}{\partial \theta} \left( \sin \theta \frac{\partial}{\partial \theta} \right) n + \frac{1}{r^2 \sin^2 \theta} \frac{\partial^2}{\partial \phi^2} n. \quad (83)$$

However, for this case, we have radial symmetry and the partial derivative terms involving  $\theta$  and  $\phi$  are equal to zero. Equation (83) therefore reduces to

$$\nabla^2 n = \left( \frac{\partial^2}{\partial r^2} + \frac{2}{r} \frac{\partial}{\partial r} \right) n, \quad (84)$$

and the neutron diffusion equation with zero Dirichlet boundary conditions becomes

$$\frac{\partial n}{\partial t} = \mu \left( \frac{\partial^2}{\partial r^2} + \frac{2}{r} \frac{\partial}{\partial r} \right) n + \eta n, \quad n = n(t, r), \quad (\mu, \eta) > 0, \quad 0 > t \geq \infty, \quad r \in \mathbb{R}; \quad (85)$$

BC:  $n(t, r_1) = 0$ ; IC:  $n(0, r) = f(r)$ ;

where  $r$  represents the radial distance from the centre of the ball,  $r_1$  the ball radius, and other symbols are as previously defined.

We use the approach of Section (2) to solve eqn. (85), where this time we postulate a solution of the form

$$n(t, r) = T(t)R(r), \quad (86)$$

so that on substituting into eqn. (85), we obtain

$$\frac{\partial TR}{\partial t} = \mu \frac{\partial^2 TR}{\partial r^2} + \mu \frac{2}{r} \frac{\partial TR}{\partial r} + \eta TR. \quad (87)$$

Dividing through by  $TR$  and rearranging, we obtain

$$\frac{1}{T} \frac{\partial T}{\partial t} - \eta = \mu \frac{1}{R} \frac{\partial^2 R}{\partial r^2} + \mu \frac{2}{rR} \frac{\partial R}{\partial r} = -\alpha, \quad (88)$$

where we again use the negated separation constant  $\alpha$  on the right hand side.

We now have two ODEs to solve, namely

$$\frac{dT}{dt} = (\eta - \alpha)T \quad \text{and} \quad \frac{d^2 R}{dr^2} + \frac{2}{r} \frac{dR}{dr} + \frac{\alpha}{\mu} R = 0. \quad (89)$$

The solution to the first of eqns. (89) is the same as we obtained for the Cartesian coordinates problems, namely

$$T = A \exp([\eta - \alpha]t); \quad (90)$$

and for a bounded solution, we must have  $\alpha > \eta$ .

The solution to the second of eqns. (89) actually has the following a closed-form analytical solution

$$R(r) = \frac{1}{r} \left[ A \sin \left( \sqrt{\frac{\alpha}{\mu}} r \right) + B \cos \left( \sqrt{\frac{\alpha}{\mu}} r \right) \right]. \quad (91)$$

However, it is required that the solution is finite at  $r = 0$ , which implies that  $B = 0$ , when the solution simplifies to

$$R(r) = \frac{A}{r} \sin \left( \sqrt{\frac{\alpha}{\mu}} r \right). \quad (92)$$

Therefore, the overall solution to eqn. (85) becomes

$$n(t, r) = \exp([\eta - \alpha]t) \frac{A}{r} \sin \left( \sqrt{\frac{\alpha}{\mu}} r \right), \quad (93)$$

where the constants have been combined into  $A$ . From the boundary conditions  $n$  must be zero at  $r = r_1$ , and this requirement will be met when the argument of the sin term is equal to  $p\pi r/r_1$ , where  $p \in \mathbb{Z}$ . But, eqn. (85) is linear therefore the *superposition principal* applies, and the solution becomes

$$n(r, t) = \sum_{p=1}^{\infty} \frac{a_p}{r} \exp([\eta - \alpha]t) \sin \left( \frac{p\pi}{r_1} r \right), \quad (94)$$

where constants  $a_p$  are determined by the initial and/or boundary conditions.

For eqn. (94) to be correct, we must have the relationship

$$\sqrt{\frac{\alpha}{\mu}} = \frac{p\pi}{r_1} \quad \Rightarrow \quad \alpha = \mu \left( \frac{p\pi}{r_1} \right)^2, \quad (95)$$

for  $p = 0, 1, 2, \dots$ . Hence, eqn. (94) becomes

$$n(r, t) = \sum_{p=1}^{\infty} \frac{a_p}{r} \exp \left( \eta t - \mu \frac{p^2 \pi^2}{r_1^2} t \right) \sin \left( \frac{p\pi}{r_1} r \right). \quad (96)$$



Note that for  $r = 0$ , when eqn. (96) becomes singular, we apply *l'Hospital's rule*<sup>2</sup> to the term  $\frac{1}{r} \sin\left(\frac{p\pi}{r_1}r\right)$ , when we obtain

$$\lim_{r \rightarrow 0} \frac{1}{r} \sin\left(\frac{p\pi}{r_1}r\right) = \frac{p\pi}{r_1}. \quad (97)$$

Now, it follows directly that for  $n(r, t)$  to increase unbounded as  $t \rightarrow \infty$ , the *critical condition* is

$$\mu \frac{p^2 \pi^2}{r_1^2} < \eta, \quad \text{or} \quad r > p\pi \sqrt{\frac{\mu}{\eta}}, \quad (98)$$

for all  $p$ . However, the worst case (smallest radius  $r$ ) is for  $p = 1$ , when the critical length is given by

$$r_{\text{crit}} = \pi \sqrt{\frac{\mu}{\eta}}. \quad (99)$$

Thus, we find that the critical volume for the for the radially symmetrical 3D case in spherical coordinates is equal to

$$V_{\text{sphere}} = \frac{4}{3}\pi \times \left(\pi \sqrt{\frac{\mu}{\eta}}\right)^3 = 129.9 \times \left(\frac{\mu}{\eta}\right)^{\frac{3}{2}}. \quad (100)$$

This compares with the cube volume for the 3D case in Cartesian coordinates of

$$V_{\text{cube}} = \left(\pi \sqrt{\frac{3\mu}{\eta}}\right)^3 = 161.1 \times \left(\frac{\mu}{\eta}\right)^{\frac{3}{2}}. \quad (101)$$

The ratio of critical volume of a cube, to critical volume of a sphere, is therefore

$$\frac{V_{\text{cube}}}{V_{\text{sphere}}} = 1.24. \quad (102)$$

Hence, to create an explosion, it takes 1.24 times as much fissile material in the shape of a cube, as would be required for fissile material in the shape of a sphere.

Using values for  $U^{235}$  from Table (7), we obtain the critical radius and volume for the 3D spherical coordinates case

$$r_{\text{crit}} = 11.04 \text{ cm}, \quad V_{\text{crit}} = 5649 \text{ cm}^3. \quad (103)$$

This is a realizable situation and corresponds to a mass of around 106 kg.

### 3.5.1 Calculate the coefficients, $a_p$

To calculate the coefficients  $a_p$  we need to set  $t = 0$  in eqn. (96) and multiply by  $r \sin\left(\frac{l\pi}{r_1}r\right)$  to obtain the correct terms on the right hand side. This is to take advantage of the orthogonality condition, see eqn. (121), when we obtain

$$\begin{aligned} \int_0^{r_1} r n(0, r) \sin\left(\frac{l\pi r}{r_1}\right) dr &= \int_0^{r_1} \sum_{p=1}^{\infty} a_p \sin\left(\frac{p\pi}{r_1}r\right) \sin\left(\frac{l\pi}{r_1}r\right) dr, \\ &= \sum_{p=1}^{\infty} \int_0^{r_1} a_p \sin\left(\frac{p\pi}{r_1}r\right) \sin\left(\frac{l\pi}{r_1}r\right) dr, \\ &= \frac{r_1}{2} a_l, \end{aligned} \quad (104)$$

---

<sup>2</sup>l'Hospital's rule is:  $\lim_{x \rightarrow a} \frac{f(x)}{g(x)} = \lim_{x \rightarrow a} \frac{f'(x)}{g'(x)}$ , where the prime indicates differentiation with respect to  $x$ . Further information can be found in any good calculus book.

Thus, exchanging  $p$  for  $\ell$  in eqn. (104), we see that the constants are given by

$$a_p = \frac{2}{r_1} \int_0^{r_1} r n(0, r) \sin\left(\frac{p\pi}{r_1} r\right) dr, \quad p = 1, 2, \dots \quad (105)$$

Setting the initial condition to  $n(0, r) = [1 - (r/r_1)^2]$ , and evaluating eqn. (105) using Maple, we obtain

$$a_p = 12 \frac{r_1 (-1)^{1+p}}{p^3 \pi^3}, \quad (106)$$


which yields the coefficients in Table (4).

**Table 4:** Calculated  $a_p$  coefficient values for the 3D spherical coordinates case.

p	ap	p	ap
1	2.1480E+00	16	-5.2440E-04
2	-2.6849E-01	17	4.3720E-04
3	7.9554E-02	18	-3.6830E-04
4	-3.3562E-02	19	3.1316E-04
5	1.7184E-02	20	-2.6849E-04
6	-9.9442E-03	21	2.3194E-04
7	6.2623E-03	22	-2.0172E-04
8	-4.1952E-03	23	1.7654E-04
9	2.9464E-03	24	-1.5538E-04
10	-2.1480E-03	25	1.3747E-04
11	1.6138E-03	26	-1.2221E-04
12	-1.2430E-03	27	1.0913E-04
13	9.7768E-04	28	-9.7848E-05
14	-7.8278E-04	29	8.8071E-05
15	6.3643E-04	30	-7.9554E-05

### 3.5.2 Maple program for 3D case - spherical coordinates and Dirichlet BC

A Maple program that calculates the coefficients,  $a_p$ , of Table (4), using parameters from Table (7), and also simulates eqn. (96), is given in Listing (5). The corresponding solution plot at time  $t = 2\text{E-}6$  [s] is provided in Figure (9). Increasing the number of terms beyond 30 makes negligible difference to the result.



**Figure 9:** Plot of  $n(t, r)$  for 3D neutron diffusion, with radial symmetry, in spherical coordinates with zero Dirichlet BC.

**Listing 5:** Maple program to calculate coefficients  $a_p$  and simulate eqn. (96) .

---

```

restart;with(Physics):
# Neutron diffusion in spherical coordinates
# Initial condition
f(r):=1-(r/r1)^2;          # IC 1
#f(r):=A*exp(-lambda*(r/r1)^2); # IC 2, Gaussian
r1:=11.5/100;A:=1;lambda:=50;
plot(f(r),r=-r1..r1,title="f(r)"); #unassign('r1');
#Calculate symbolic solution for a[p]
eqn3:=(2/r1)*Int(r*f(r)*sin((p)*Pi*r/r1),r=0..r1) =
      (2/r1)*int(r*f(r)*sin((p)*Pi*r/r1),r=0..r1) assuming(p,integer);
unassign('p');
evalf(subs(p=1,rhs(eqn3)));
# Calculate values for a[p]
N:=30;lambda:=100;
for p from 1 by 1 to N do
    aa[p]:=evalf(rhs(eqn3)); # print(Re(aa[p]));
end do: unassign('p');
# Calculate solution
mu:=2.3446e5;eta:=1.8958e8;
rcrit:=evalf(Pi*sqrt(mu/eta));
n:= sum((aa[p]/r)*exp(eta*t-mu*((p)*Pi/r1)^2*t)*sin((p)*Pi*r/r1),p=1..N):
# Plot solution surface
plot3d([r,t,n],r=-r1..r1,t=0..2e-6,axes=framed,title="Neutron diffusion,
      r=11.5cm, N=30",labels=["r", "t", "n(t=2e-6,r)"],orientation=[-68, 70, 0],
      grid=[100,100]); #orientation=[-90, 90, 0]
plot(f(r)-subs({t=0},n),r=-r1..r1);
#plot(f(r),r=-r1..r1);
#plot(subs({t=0},n),r=-r1..r1);

```

---

### 3.6 3D case in spherical coordinates - Neumann boundary condition

We now consider the same problem as outlined in section (3.5), but with Neumann boundary conditions, when the neutron diffusion equation becomes

$$\begin{aligned} \frac{\partial n}{\partial t} &= \mu \left( \frac{\partial^2}{\partial r^2} + \frac{2}{r} \frac{\partial}{\partial r} \right) n + \eta n, \quad n = n(t, r), \quad (\mu, \eta) > 0, \quad 0 > t \geq \infty, \quad r \in \mathbb{R}; \\ \text{BC: } \frac{dn(t, r_1)}{dr} &= -\frac{3}{2} \frac{n(t, r_1)}{\lambda_t}; \quad \text{IC: } n(0, r) = f(r); \end{aligned} \quad (107)$$

where  $r$  represents the radial distance from the centre of the ball,  $r_1$  the ball radius,  $\lambda_t$  represents transport free path. Recall that the neutron rate of formation constant is equal to  $\eta = v_{neut}(\nu - 1)/\lambda_f$ , where  $\nu$  represents *secondary neutrons created by fission* (the  $-1$  accounts for the neutron causing fission being consumed in the reaction). Other symbols are as previously defined. For discussion on the derivation of eqn. (107) and the boundary condition, see [Ree-15, Appendix G] and [Fey-64, vol. II, Section 12-4].

We use the approach of Section (3.5) to solve eqn. (107), where we again postulate a solution of the form

$$n(t, r) = T(t)R(r), \quad (108)$$

so that on substituting into eqn. (85), we obtain

$$\frac{\partial TR}{\partial t} = \mu \frac{\partial^2 TR}{\partial r^2} + \mu \frac{2}{r} \frac{\partial TR}{\partial r} + \eta TR = -\alpha, \quad (109)$$

where we again use the negated separation constant  $\alpha$  on the right hand side.

Proceeding as previously, we obtain the following two ODEs,

$$\frac{dT}{dt} = -\alpha T \quad \text{and} \quad \frac{d^2 R}{dr^2} + \frac{2}{r} \frac{dR}{dr} + \frac{\eta + \alpha}{\mu} R = 0, \quad (110)$$

except that in this instance the variable  $\eta$  is assigned to the spatial ODE.

The solution to the first of eqns. (110) is

$$T = A \exp(-\alpha t); \quad (111)$$

and for a bounded solution, we must have  $\alpha > 0$ .

The solution to the second of eqns. (110) again has the following a closed-form analytical solution

$$R = \frac{1}{r} [B_1 \sin(kr) + B_2 \cos(kr)], \quad (112)$$

where  $k = \sqrt{\frac{\eta + \alpha}{\mu}}$

However, it is required that the solution is finite at  $r = 0$ , which implies that  $B_2 = 0$ , when the solution simplifies to

$$R = \frac{B_1}{r} \sin(kr). \quad (113)$$

We now impose the Neumann boundary condition

$$\frac{dR}{dr} = -\frac{3}{2} \frac{R}{\lambda_t}, \quad (114)$$

onto eqn(110), which yields

$$-\frac{B_1 \sin(kr)^2}{r} + \frac{B_1 \cos(kr)k}{r} = -\frac{3}{2} \frac{B_1 \sin(kr)}{r \lambda_t}. \quad (115)$$

Rearranging this equation we obtain

$$-1 + kr \cot(kr) + \frac{3}{2} \frac{r}{\lambda_t} = 0. \quad (116)$$

If we let  $\alpha = 0$ , this will create the *criticality condition*, where  $n$  is neither increasing nor decreasing, and eqn (116) becomes

$$-1 + r \sqrt{\frac{\eta}{\mu}} \cot \left( r \sqrt{\frac{\eta}{\mu}} \right) + \frac{3}{2} \frac{r}{\lambda_t} = 0. \quad (117)$$

We are now in a position to calculate a value for the critical radius of the device by solving eqn. (117) for  $r$  using values from Table (7). Using Maple we obtain the following critical values


$$r_{\text{crit}} = 8.369 \text{ cm}, \quad V_{\text{crit}} = 2456 \text{ cm}^3. \quad (118)$$

This is a realizable situation and corresponds to a mass of around 45.9 kg.

Thus, using the more realistic Neumann BC of eqn. (114), we obtain a much smaller critical spherical mass than was predicted in Section (3.5), using a zero Dirichlet BC.

If we now choose the sphere to be slightly larger than the critical sphere, say  $R = 8.5$  cm, we can back calculate the corresponding separation constant, which we find to be  $\alpha = -4.401\text{E}7$ ; and this yields  $k = 0.2809$ .

Therefore, the solution to eqn. (107) is completely known, except for constants  $A$  and  $B_1$  which we combine into  $A$ . To evaluate the constant  $A$  we need to impose a suitable initial condition. The exact form of the IC,  $f(r)$ , is not important, because  $r > r_{\text{crit}}$  and thus any value will cause  $n(r, 0)$  to increase exponentially with time. We therefore choose to set  $n = 1$  at the surface of the device, i.e.  $n(r = 8.5, t = 0) = 1$ , which makes  $A = 12.415$ . A plot of the radially symmetric IC is shown in Figure (10).



**Figure 10:** Plot of radially symmetric initial condition for 3D neutron diffusion in spherical coordinates and Neumann BC, with  $r = 8.5$  cm.


Thus, all the parameters are now known and the final solution to eqn. (107) becomes,

$$n(t, r) = A \exp(-\alpha t) \frac{\sin(kr)}{r}. \quad (119)$$

### 3.6.1 Maple program for 3D case - spherical coordinates and Neumann BC

A Maple program that calculates the solution of eqn (107) using parameters from Table (7), and also simulates eqn. (119), is given in Listing (6).

The corresponding plot of the solution to time  $t = 3\text{E}-6$  [s] is provided in Figure (11).



**Figure 11:** Plot of  $n(t, r)$  for 3D neutron diffusion, with radial symmetry, in spherical coordinates with  $r = 8.5$  cm. Note how the Neumann BC is enforced, and that the value of  $n(r = \pm 8.5, t)$  increases with time, unlike the case with Dirichlet BC.

**Listing 6:** Maple program to calculate parameters  $\alpha$  and  $A$ , and that also simulates eqn. (119).

```
# 3D neutron diffusion in spherical coordinates - Neumann BC
restart;with(DEtools):with(plots):
pde := diff(u(r,t),t) = mu*(diff(u(r,t),r,r)+(2/r)*diff(u(r,t),r))+eta*u(r,t);
pde2:=algsb(u(r,t)=R(r)*T(t),pde)/(R(r)*T(t));
# Variable separable - alternative approach
ode1:=(mu*(diff(R(r), r, r))*r+2*mu*(diff(R(r), r))+eta*R(r)*r)/(R(r)*r)+alpha
=0;
ode2 := diff(R(r),r,r) + 2*(diff(R(r),r))/r + R(r)*k^2; # k^2 = (eta+alpha)/mu
BC:=D(R)(0)=0;
solR:=dsolve({ode2,BC}) assuming(eta>0,alpha>0); assign(solR);
# Neumann BC # Reed p59
eqn1:=diff(R(r),r)= - R(r)*3/(2*lambda[t]);
# Manually simplify
eqn2:= -1 + k*r*cot(k*r)+(3/2)*r/lambda[t];
k:=sqrt((eta+alpha)/mu);
# Data U235, Reed p 56
tau:=8.6349e-9; # [s]
lambda[f]:=0.1689*100; # [cm]
lambda[t]:=0.0360*100; # [cm]
nu:=2.637; # [1/cm^3]
rho:=18.71; # [g/cm^3]
v:=lambda[f]/tau; # [cm/s]
mu:=lambda[t]*v/3; # [cm^2/s]
eta:=v*(nu-1)/lambda[f]; # [1/s]
# set alpha:=0 - criticality condition: N neither increasing nor decreasing
Rcrit:=fsolve(subs(alpha=0,eqn2),r); # [cm]
Vcrit:= evalf((4/3)*Pi*Rcrit^3); # [cm^3]
Mcrit:=evalf(rho*Vcrit); # [grams]
# Set r > Rcrit to get runaway reaction
R0:=8.5; # [cm]
alpha_0:=fsolve(subs(r=R0,eqn2),alpha);
alpha:=alpha_0;
sprintf("k=%f",k);
N(r,t):= A*exp(-alpha_0*t)*sin(k*r)/r; # constant A includes _C1
eqn3:=subs({r=R0,t=0},N(r,t))=1;
solA:=isolate(eqn3,A);
AA := rhs(solA);
N(r,t):=subs(A=AA,N(r,t));
plot(subs(t=0,N(r,t)),r=-R0..R0,axes=framed,title="Neutron diffusion, r=8.5cm,
```



```
t=0", labels=["r", "n(r,t=0)"], numpoints=1000, thickness=2); #, gridlines=true);
plot3d(simplify(N(r,t), symbolic), r=-R0..R0, t=0..3e-6, axes=framed,
title="Neutron diffusion, r=8.5cm", labels=["r", "t", "n(r,t=3e-6)"],
orientation=[-68, 70, 0], grid=[100,100]);
```

---

## 4 Determining eigenfunction coefficients

Recall from trigonometry that for any real  $a$  or  $b$  the following identity holds,

$$\sin(a) \sin(b) = \frac{1}{2} [\cos(a - b) - \cos(a + b)]. \quad (120)$$

It therefore follows, because functions  $\sin(k\pi x/L)$  and  $\sin(\ell\pi x/L)$  are orthogonal that

$$\int_0^L \sin\left(\frac{k\pi x}{L}\right) \sin\left(\frac{\ell\pi x}{L}\right) dx = \begin{cases} 0 & k \neq \ell, \\ L/2 & k = \ell \end{cases}. \quad (121)$$

Now consider the 1D eigenfunction solution given by eqn. (18). When evaluated at  $t = 0$  the exponential term is equal to unity and we have  $n(0, x) = f(x)$  or,

$$f(x) = \sum_{p=1}^{\infty} a_p \sin\left(\frac{p\pi}{L}x\right). \quad (122)$$

It therefore follows from eqn. (121) that

$$\begin{aligned} \int_0^L f(x) \sin\left(\frac{\ell\pi x}{L}\right) dx &= \int_0^L \sum_{p=1}^{\infty} a_p \sin\left(\frac{p\pi}{L}x\right) \sin\left(\frac{\ell\pi}{L}x\right) dx, \\ &= \sum_{p=1}^{\infty} \int_0^L a_p \sin\left(\frac{p\pi}{L}x\right) \sin\left(\frac{\ell\pi}{L}x\right) dx, \\ &= \frac{L}{2} a_\ell, \end{aligned} \quad (123)$$

Thus, the constants are given by

$$a_p = \frac{2}{L} \int_0^L f(x) \sin\left(\frac{p\pi}{L}x\right) dx, \quad p = 1, 2, \dots. \quad (124)$$

Note that if we had taken a double-sided summation from  $-\infty$  to  $+\infty$  in eqn. (18), the right hand side of eqn. (123) would be equal to  $La_\ell$ . This approach would result in  $|a_p|$  being equal to half the value of the single-sided summation  $a_p$ , with  $a_{-p} = -a_p$ . However,  $-p$  in the sin terms means that summation terms for  $-p$  and  $+p$  are both the same sign and are therefore additive. Consequently, the double-sided summation requires twice the number of computations to achieve the same accuracy as a single-sided summation. For this reason, we use the single-sided summation in eqn. (18).

On applying the above to the 1D problem of section (3.1), with the following Gaussian initial condition,

$$f(x) = A \exp\left(\frac{-4\lambda \left(x - \frac{1}{2}L\right)^2}{L^2}\right) \quad (125)$$

where  $A = 1$ ,  $\lambda = 100$  and  $L = 11.1$ , we obtain the  $a_p$  constants (up to  $p = 29$ ) given in Table (5). We also used values  $\eta = 2.3446\text{E}05$  and  $\mu = 1.8958\text{E}08$  for  $\text{U}^{235}$  from Table

**Table 5:** Coefficient values for 1D neutron diffusion problem. Note that values for even  $p$  are equal to zero.

$p$	$a_p$	$p$	$a_p$
1	0.1762	17	0.0298
3	-0.1677	19	-0.0191
5	0.1519	21	0.0117
7	-0.1310	23	-0.0068
9	0.1075	25	0.0038
11	-0.0840	27	-0.0020
13	0.0625	29	0.0010
15	-0.0442		

(7). Thus, from eqn. (22), we see that  $L_{\text{crit}} = 11.05$ , which means that the situation is super-critical ( $L > L_{\text{crit}}$ ), and therefore  $n$  should increase exponentially with time.

For problems defined in cylindrical coordinates, variable separable solutions also generate series solutions. However, they usually require the use of Bessel functions to generate the associated coefficients. See appendix (A) for more details.

## 5 The addition of a tamper

The analysis up until now has considered only the fissile material with a Dirichlet BC of the form  $n(t, r = L) = 0$ , or a Neumann BC of the form  $\frac{d}{dt}n(t, r = r_1) = f(n(t, r = r_1))$ . These are not the most efficient ways of creating a critical chain reaction. A more efficient way is to surround the fissile material with appropriate *non-fissile* material, which leads to a situation where some neutrons are reflected back into the fissile material, rather than escaping to the atmosphere. This has the effect of retarding the escape of neutrons and results in a system where a critical chain reaction occurs with a much smaller critical mass. This is important as fissile material is expensive, so has to be used as efficiently as possible. The non-fissile material is called a *tamper*<sup>3</sup> (more recently termed a *reflector* or *containment shell*). Also, due to its inertia, the tamper helps to retard the expansion of the active material, thereby prolonging the chain reaction, which further improves explosion efficiency. Note: retardation due to the tamper's tensile strength is negligible. It has been found that heavy elements have better neutron reflecting properties than lighter elements.

The addition of a tamper means that the PDE described by eqn. (9) now has to also describe this extra material. However, we are considering a tamper made of non-fissile material, so it is subject only to diffusion, i.e. there is no neutron source. Thus, the domain  $\Omega$  increases to include the tamper, and  $\eta = 0$  and  $f(x) = 0$  within the tamper. Also, the diffusion constant  $\mu$  within the tamper may be different to that of the fissile material. The conditions on the new boundary will still be of the Dirichlet type, with  $n = 0$  on  $\partial\Omega$ . The effect of all this is that some exiting neutrons on the fissile material/tamper boundary will be reflected back, so speeding up the nuclear reaction, leading to an explosion of greater intensity. However, adding a tamper makes solving the associated diffusion equation more difficult as closed form solutions, such as the series solutions found above, are not known. Therefore, a solution has to be obtained by numerical methods.

<sup>3</sup>The term tamper originally derived from analogy to the material - usually clay - tamped around commercial explosives.

## 6 Neutron mean free path

The *mean free path* is equal to the average distance traveled by a neutron without any interaction - also known as the *characteristic length*.

Consider the situation where a thin slab of material of thickness  $s$  and cross-sectional area  $\Sigma$  is bombarded by incoming neutrons at a rate  $R_0$  neutrons/(m<sup>2</sup>s). Under these conditions the number density of nuclei in the material is given by [Ree-15, section (2.1)]

$$n = 10^6 \left( \frac{\rho N_A}{A} \right), \quad (126)$$

where  $\rho$  represents bulk density,  $N_A$  Avogadro's number and  $A$  atomic weight.

We wish to calculate how many reactions will occur in unit time as a result of the neutron bombardment rate  $R_0$ . The volume of the slab is  $\Sigma s$  which means that it will contain  $\Sigma s n$  nuclei. Assume that each nucleus presents a reaction cross-sectional area of  $\sigma$  to the incoming neutrons. Then the total area represented by all the nuclei in the slab will be  $\Sigma s n \sigma$ , and the fraction of the slab's surface area available for reactions to occur is equal to  $\Sigma s n \sigma / \Sigma = s n \sigma$ . Therefore, the rate of reactions is given by

$$R = (R_0 \Sigma) s n \sigma. \quad (127)$$

It also follows that the probability that an individual neutron precipitates a reaction is equal to the reaction rate  $R$  divided by the incident neutron flux  $R_0 \Sigma$ , i.e.

$$P_{rx} = s n \sigma. \quad (128)$$

Consequently, the probability of a neutron passing through the slab and escaping, will be

$$P_{esc} = 1 - P_{rx} = 1 - s n \sigma. \quad (129)$$

Now consider a block of thickness  $x$  made up of  $m$  slabs each of thickness  $s$  so that we have  $x = m s$ . If  $N_0$  neutrons are incident on the face of the first block, then the number that would pass through the first slab would be  $N_0 P_{esc}$ . These neutrons would then be incident on the second slab, and the number that would pass through it would be  $N_0 P_{esc}^2$ , and so on. Thus, the number of neutrons that pass through the entire block would be  $N_{esc} = N_0 P_{esc}^m$ . Alternatively, we may write

$$N_{esc} = N_0 (1 - s n \sigma)^{x/s}. \quad (130)$$

We can rearrange this equation to give

$$N_{esc} = N_0 (1 + z)^{-s n \sigma / z} = N_0 \left[ (1 + z)^{1/z} \right]^{-\sigma n x}, \quad (131)$$

where  $z = -s n \sigma$ . But  $s$  is very small and we let  $z \rightarrow 0$ . Therefore the number of neutrons that pass through the block to a distance  $x$  becomes<sup>4</sup>

$$N_{esc} = N_0 e^{-\sigma n x}. \quad (132)$$

Following similar arguments, the number of neutrons that initiate a reaction is given by

$$N_{rx} = 1 - N_{esc} = N_0 (1 - e^{-\sigma n x}). \quad (133)$$

---

<sup>4</sup>The definition of the base of *natural logarithms* is,  $e := \lim_{n \rightarrow \infty} (1 + 1/n)^n$ , which is equivalent to  $\lim_{z \rightarrow 0} (1 + z)^{1/z}$ .

It therefore follows that the probability of a single neutron initiating a reaction in the block becomes

$$P_{rx} = \frac{N_{rx}}{N_0} = (1 - e^{-\sigma n x}), \quad (134)$$

with a corresponding *probability density function*<sup>5</sup>

$$p_{rx} = \frac{d}{dx} (P_{rx}) = \sigma n e^{-\sigma n x}. \quad (135)$$

We are now in a position to calculate the neutron mean free path, which is equal to the *mean value* or *expectation* of  $x$ , i.e.

$$x_{mfp} = E(x) = \int_0^L x p_{rx} dx. \quad (136)$$

From eqn. (135) this becomes

$$x_{mfp} = \frac{1 - e^{-\sigma n L} - e^{-\sigma n L} \sigma n L}{\sigma n}, \quad (137)$$

which, for large  $L$ , reduces to

$$x_{mfp} = \frac{1}{\sigma n}. \quad (138)$$

Using fission parameters from Table (7), we finally obtain the following value for the neutron mean free path of  $U^{235}$ ,

$$x_{mfp} = 16.89 \text{ cm}. \quad (139)$$

A Maple program that performs the above for the neutron free path calculations is given in Listing (7).

**Listing 7:** Maple program to calculate the neutron mean free path.

---

```
# Mean free path calculation for neutrons bombarding a slab
restart;
# Ref: Reed, section 2.1
# Slab calculations
#####
# Reactions per second
# R[0] = rate of neutron bombardment (neutrons/(m^2 s))
# sigma = effective cross-sectional area of nucleus
# Sigma = cross-sectional area (m^2)
# n = number density of nuclei (1/m^3)
# s = depth of slab (m)
R[N]:=R[0]*Sigma*s*n*sigma; # (1/s)
# Probability of a reaction
P[react]:=R[N]/(R[0]*Sigma); # (-)
# Probability of escape
P[escape]:=1-P[react]; # (-)
# Block calculations
#####
# Neutrons that emerge from block made up of m slabs
# x = depth of block (m)
# N[0] = number of neutrons incident on block
m := x/s;
eqn1:=N[escape]=N[0]*P^m;
N[escape]:=subs(P=P[escape],rhs(eqn1));
restart;
```

---

<sup>5</sup>For a continuous *cumulative distribution function*  $F(x)$ , the probability density function is defined as  $f(x) = \frac{d}{dx} (F(x))$ .

```

# Limit as z -> 0
#####
# Define z = -s*n*sigma, then we have
eqn2:=N[escape]=N[0]*(1+z)^(-sigma*n*x/z);
# therefore
N[escape]:=N[0]*[(1+z)^(1/z)]^(-sigma*n*x);
# Let z -> 0
k:=limit((1+z)^(1/z),z=0);
N[escape]:=N[0]*exp(-sigma*n*x);
N[react]:=N[0]-N[escape];

# Therefore, probability a neutron will travel a distance x is,
P[directEscape]:= N[escape]/N[0];
P[react]:= 1-N[escape]/N[0];
# Probability density function - react
#####
p[react]:= diff(P[react],x);
# Mean value of x
#####
xm:=simplify(int(x*p[react],x=0..L), 'symbolic' ) assuming sigma::positive,n::
positive;
# Number density
#####
n:=10^6*(rho*N[A]/At); # (1/m^3)
# Data for U235
rho := 18.71;          # (g/cm^3) Bulk density
N[A] := 6.022e23;      # (1/mole) Advogadro's Number
At := 235.04;          # (g/mole) Atomic weight
sigma:= 1.235e-28;     # (m^2) Neutron cross-sectional area
# Neutron mean free path
#####
x[meanFreePath]:=limit(xm,L=infinity); # (m)
plot(xm,L=0..2);
plot(P[directEscape],x=0..2);
plot(P[react],x=0..2);

```

---

## Appendices

### A Orthogonality of Bessel functions

In Fourier theory we have the *orthogonality condition* for sine functions

$$\int_0^L \sin\left(\frac{m\pi x}{L}\right) \sin\left(\frac{n\pi x}{L}\right) dx = \begin{cases} 0 & m \neq n, \\ L/2 & m = n \end{cases}, \quad (140)$$

where the sine terms are equal to zero at the integration limits. There is also a similar orthogonal condition for Bessel functions.

Here we discuss Bessel functions of the first kind, that have integer order, and that are finite at the origin. These are also known as *cylindrical functions* or *cylindrical harmonics*. This is because they are employed in generating solutions to problems that are formulated in cylindrical coordinates.

Consider the following  $m$ th-order Bessel functions

$$\begin{aligned} x^2 \frac{d^2 u}{dx^2} + x \frac{du}{dx} + (a^2 x^2 - m^2) u &= 0, & u &= u(x), & m &\in \mathbb{Z}, \\ x^2 \frac{d^2 v}{dx^2} + x \frac{dv}{dx} + (b^2 x^2 - m^2) v &= 0, & v &= v(x), & m &\in \mathbb{Z}, \end{aligned} \quad (141)$$

which have, respectively, solutions

$$J_m(ax) \quad \text{and} \quad J_m(bx), \quad (142)$$

where  $a$  and  $b$  each represent a different *zero* of  $J_m(x)$ . Thus, it follows that the functions of eqns. (142) are equal to zero when  $x = 1$ .

If we multiply the first equation by  $v$  and the second by  $u$ , on subtracting and dividing through by  $x$ , we obtain

$$(b^2 - a^2) xuv = \frac{d}{dx} \left( vx \frac{du}{dx} - ux \frac{dv}{dx} \right). \quad (143)$$

Then, integration between the limits of 0 and 1 yields

$$(b^2 - a^2) \int_0^1 xuv dx = \left( vx \frac{du}{dx} - ux \frac{dv}{dx} \right) \Big|_0^1 = 0. \quad (144)$$

Recall that  $u = J_m(ax)$  and  $v = J_m(bx)$ , and that  $J_m(ax)|_{x=1} = 0$  and  $J_m(bx)|_{x=1} = 0$ . Thus, it follows that the right hand side of eqn. (144) is equal to zero at  $x = 0$  and at  $x = 1$ . Therefore, we conclude that for  $a \neq b$ , the following identity holds

$$\int_0^1 x J_m(ax) J_m(bx) dx = 0, \quad (145)$$

which is the Bessel function orthogonality condition. Note that for Bessel functions, the  $x$  term in the integrand of eqn. (145) represents a *weight function*.

For the situation where  $a = b$ , it can be shown that

$$\int_0^1 x J_m^2(ax) dx = \frac{1}{2} J_{m+1}^2(a). \quad (146)$$

These results enable the coefficients in a series expansion of a function to be determined in terms of Bessel functions. Given a function  $f(x)$ , the Bessel function expansion in the range  $0 < x < x_0$  becomes

$$f(x) = \sum_{n=0}^{\infty} c_n J_m \left( \frac{\alpha_n}{x_0} x \right), \quad (147)$$

where  $\alpha_n$  are zeros of  $J_m(x)$  such that  $J_m \left( \frac{\alpha_n}{x_0} x \right) \Big|_{x=x_0} = 0$ .

It therefore follows from eqns. (145) and (147) that

$$\begin{aligned} \int_0^{x_0} f(x) x J_m \left( \frac{\alpha_n}{x_0} x \right) dx &= \int_0^{x_0} \sum_{n=0}^{\infty} c_n J_m \left( \frac{\alpha_n}{x_0} x \right) x J_m \left( \frac{\alpha_q}{x_0} x \right) dx, \\ &= \sum_{n=1}^{\infty} \int_0^{x_0} c_n J_m \left( \frac{\alpha_n}{x_0} x \right) x J_m \left( \frac{\alpha_q}{x_0} x \right) dx, \\ &= \begin{cases} \frac{a}{2} c_q J_{m+1}^2 \left( \frac{\alpha_q}{x_0} x \right) & , \quad n = q, \\ 0 & , \quad n \neq q. \end{cases} \end{aligned} \quad (148)$$

Thus, the constants  $c_q$  in the expansion are given by

$$c_q = \frac{\int_0^{x_0} f(x) x J_m \left( \frac{\alpha_q}{x_0} x \right) dx}{\frac{x_0^2}{2} J_{m+1}^2 \left( \frac{\alpha_q}{x_0} x \right)}, \quad q = 1, 2, \dots \quad (149)$$

A good introduction to Bessel functions may be found in [Asm-05].



## B Bessel function zeros

**Table 6:** The first 30 *zeros* of Bessel function  $J_0(x)$

$q$	1	2	3	4	5	6
$\alpha_q$	2.404826	5.520078	8.653728	11.79153	14.93092	18.07106
$q$	7	8	9	10	11	12
$\alpha_q$	21.21164	24.35247	27.49348	30.63461	33.77582	36.9171
$q$	13	14	15	16	17	18
$\alpha_q$	40.05843	43.19979	46.34119	49.48261	52.62405	55.76551
$q$	19	20	21	22	23	24
$\alpha_q$	58.90698	62.04847	65.18996	68.33147	71.47298	74.6145
$q$	25	26	27	28	29	30
$\alpha_q$	77.75603	80.89756	84.03909	87.18063	90.32217	93.46372

## C Physical parameters for $U^{235}$ and $Pu^{239}$

Some useful data, derived from [Ree-15], for fissile materials  $U^{235}$  and  $Pu^{239}$  are given in Table (7)

**Table 7:** Some parameters for  $U^{235}$  and  $Pu^{239}$ . Adapted from [Ree-15].

symbol	$U^{235}$	$Pu^{239}$	units	description
$\sigma_f$	1.2350E-28	1.8000E-28	m <sup>2</sup>	fission cross-section
$\sigma_{el}$	4.5660E-28	4.3940E-28	m <sup>2</sup>	elastic scattering cross-section
$\sigma_t$	5.8010E-28	6.1940E-28	m <sup>2</sup>	total/transport cross-section, $(\sigma_f + \sigma_{el})$
$n$	4.7940E+28	3.9300E+28	1/m <sup>3</sup>	neutron number density
$\lambda_f$	0.1689	0.1414	m	fission free path, $1/(n\sigma_f)$
$\lambda_t$	0.0360	0.0411	m	transport free path, $1/(n\sigma_t)$
$\nu$	2.6370	3.1720	-	secondary neutrons created by fission
$\rho$	18.7100	15.6000	g/cm <sup>3</sup>	density
$A_t$	235.04	239.05	-	atomic weight
$E_N$	2	2	MeV	neutron energy
$e_p$	1.6022E-13	1.6022E-13	J/MeV	conversion factor
$m_N$	1.6749E-27	1.6749E-27	kg	neutron mass
$v$	1.9561E+07	1.9561E+07	m/s	neutron speed, $\sqrt{2e_p E_N/m_N}$
$\mu$	2.3446E+5	2.6786E+5	m <sup>2</sup> /s	diffusion constant, $\lambda_t v/3$
$\eta$	1.8958E+8	3.0055E+8	1/s	neutron generation rate, $v(\nu - 1)/\lambda_f$
$\tau$	8.6347E-9	7.2268E-9	s	mean travel time to neutron fission, $\lambda_f/v$

## D Summary of critical conditions

The  $U^{235}$  critical conditions (*length, radius, volume, mass*) for the various *untamped* cases discussed above are given in Table (8). From these results we see that the 3D spherical coordinates case has the lowest critical volume, and the ratios are

$$V_{\text{sphere}} : V_{\text{cyl}} : V_{\text{cube}} = 1 : 1.142 : 1.241$$

for spherical, Cartesian and cylindrical coordinates respectively.

**Table 8:** Summary of critical conditions for  $^{235}\text{U}$ . Critical masses for Dirichlet BC cases will be smaller owing to actual boundary conditions being of the Neumann type, see [Ree-15, Section 2.2 and Appendix G].

Case	$L_{\text{crit}}$ [cm]	$r_{\text{crit}}$ [cm]	$V_{\text{crit}}$ [cm <sup>3</sup> ]	$M_{\text{crit}}$ [kg]
1D, Cartesian*	11.05	-	-	-
2D, Cartesian*	15.62	-	-	-
3D, Cartesian*	19.14	-	7007	131.1
3D, cylindrical*	19.14	10.36	6450	120.7
3D, spherical*	-	11.05	5649	105.7
3D, spherical <sup>†</sup>	-	8.369	2456	45.9

\* = Dirichlet BC,    † = Neumann BC

The above values are in agreement with those given in [Ree-15].

## References

- [Asm-05] **Asmar, N. H.** (2005). *Partial Differential Equations: with Fourier series and boundary Value problems*, (2nd ed.), New Jersey, USA: Pearson International.
- [Fey-64] **Feynman, R., R. B. Leighton and M. L. Sands** (1964), *The Feynman Lectures on Physics*, Addison-Wesley Publishing Co.
- [Ree-15] **Reed, B. C.** (2015), *The Physics of the Manhattan project*, 3rd Ed., Springer.
- [Rho-86] **Rhodes, R.** (1986), *The Making of the Atomic Bomb*, Simon and Schuster.
- [Ser-92] **Serber, R.** (1992), The Los Alamos primer, the first lectures on how to build an atomic bomb. Annotated by Robert Serber. Edited with an introduction by Richard Rhodes, *University of California Press*.